



HAL
open science

From Tarski to Hilbert

Gabriel Braun, Julien Narboux

► **To cite this version:**

Gabriel Braun, Julien Narboux. From Tarski to Hilbert. Automated Deduction in Geometry 2012, Jacques Fleuriot, Sep 2012, Edinburgh, United Kingdom. pp.89-109, 10.1007/978-3-642-40672-0_7. hal-00727117v2

HAL Id: hal-00727117

<https://inria.hal.science/hal-00727117v2>

Submitted on 22 Mar 2013

HAL is a multi-disciplinary open access archive for the deposit and dissemination of scientific research documents, whether they are published or not. The documents may come from teaching and research institutions in France or abroad, or from public or private research centers.

L'archive ouverte pluridisciplinaire **HAL**, est destinée au dépôt et à la diffusion de documents scientifiques de niveau recherche, publiés ou non, émanant des établissements d'enseignement et de recherche français ou étrangers, des laboratoires publics ou privés.

From Tarski to Hilbert

Gabriel Braun and Julien Narboux

ICube, UMR 7357 University of Strasbourg - CNRS

Abstract. In this paper, we report on the formal proof that Hilbert’s axiom system can be derived from Tarski’s system. For this purpose we mechanized the proofs of the first twelve chapters of Schwabäuser, Szmielew and Tarski’s book: *Metamathematische Methoden in der Geometrie*. The proofs are checked formally within classical logic using the Coq proof assistant. The goal of this development is to provide clear foundations for other formalizations of geometry and implementations of decision procedures.

1 Introduction

Euclid is considered as the pioneer of the axiomatic method. In the *Elements*, starting from a small number of self-evident truths, called postulates or common notions, he derives by purely logical rules most of the geometrical facts that were discovered in the two or three centuries before him. But upon a closer reading of Euclid’s *Elements*, we find that he does not adhere as strictly as he should to the axiomatic method. Indeed, at some steps in some proofs he uses a method of “superposition of triangles”. This kind of justification cannot be derived from his set of postulates¹.

In 1899, in *der Grundlagen der Geometrie*, Hilbert described a more formal approach and proposed a new axiom system to fill the gaps in Euclid’s system.

Recently, the task consisting in mechanizing Hilbert’s *Grundlagen der Geometrie* has been partially achieved. A first formalization using the Coq proof assistant [2] was proposed by Christophe Dehlinger, Jean-François Dufour and Pascal Schreck [3]. This first approach was realized in an intuitionist setting, and concluded that the decidability of point equality and collinearity is necessary to check Hilbert’s proofs. Another formalization using the Isabelle/Isar proof assistant [4] was performed by Jacques Fleuriot and Laura Meikle [5]. Both formalizations have concluded that, even if Hilbert has done some pioneering work about formal systems, his proofs are in fact not fully formal, in particular degenerated cases are often implicit in the presentation of Hilbert. The proofs can be made more rigorous by machine assistance. Indeed, in the different editions of *die Grundlagen der Geometrie* the axioms were changed, but the proofs were not always changed accordingly, this obviously resulted in some inconsistencies. The use of a proof assistant solves this problem: when an

¹ Recently, Jeremy Avigad and Edward Dean and John Mumma have shown that it is possible to define a formal system to model the proofs of Euclid’s *Elements* [1]

axiom is changed it is easy to check if the proofs are still valid. In [6], Phil Scott and Jacques Fleuriot proposed a tool to write readable formalised proof-scripts that correspond to Hilbert’s prose arguments.

In the early 60s, Wanda Szmielew and Alfred Tarski started the project of a treaty about the foundations of geometry based on another axiom system for geometry designed by Tarski in the 20s². A systematic development of Euclidean geometry was supposed to constitute the first part but the early death of Wanda Szmielew put an end to this project. Finally, Wolfram Schwabhäuser continued the project of Wanda Szmielew and Alfred Tarski. He published the treaty in 1983 in German: *Metamathematische Methoden in der Geometrie* [8]. In [9], Art Quaife used a general purpose theorem prover to automate the proof of some lemmas in Tarski’s geometry, but the lemmas which can be solved using this technique are some simple lemmas which can be proved within Coq using the `auto` tactic. The axiom system of Tarski is quite simple and has good meta-theoretical properties. Tarski’s axiomatization has no primitive objects other than points. This allows us to change the dimension of the geometric space without changing the language of the theory (whereas in Hilbert’s system one needs the notion of ‘plane’). Some axioms provide a means to define the lower and upper dimension of the geometric space. Gupta proved the axioms independent [10], except the axiom of Pasch and the reflexivity of congruence (which remain open problems).

In this paper we describe our formalization of the first twelve chapters of the book of Wolfram Schwabhäuser, Wanda Szmielew and Alfred Tarski in the Coq proof assistant. Then we answer an open question in [5]: Hilbert’s axioms can be derived from Tarski’s axioms and we give a mechanized proof. Alfred Tarski worked on the axiomatization and meta-mathematics of euclidean geometry from 1926 until his death in 1983. Several axiom systems were produced by Tarski and his students. In this formalization, we use the version presented in [8].

We aim at one application: the use of a proof assistant in education to teach geometry [11]

This theme has already been partially addressed by the community. Frédérique Guilhot has realized a large Coq development about Euclidean geometry following a presentation suitable for use in french high-school [12] and Tuan-Minh Pham has further improved this development [13]. We have presented the formalization and implementation in the Coq proof assistant of the area decision procedure of Chou, Gao and Zhang [17–20] and of Wu’s method [22, 21].

Formalizing geometry in a proof assistant has not only the advantage of providing a very high level of confidence in the proof generated, it also permits us to insert purely geometric arguments within other kind of proofs such as, for instance, proof of correctness of programs or proofs by induction. But for the time being most of the formal developments we have cited are distinct and as they do not use the same axiomatic system, they cannot be combined. In [23], we have shown how to prove the axioms of the area method within the formalization of geometry by Guilhot and Pham.

² These historical pieces of information are taken from the introduction of the publication by Givant in 1999 [7] of a letter from Tarski to Schwabhäuser (1978).

The goal of our mechanization is to do another step toward the merging of all these developments. We aim at providing very clear foundations for other formalizations of geometry and implementations of decision procedures.

We will first describe the axiom system of Tarski, its formalization within the Coq proof assistant. As our other Coq developments about geometry are in 2D, we limit ourselves to 2-dimensional geometry. Then we give a quick overview of the formalization. To show the difficulty of the task, we will give the proof of one of the non trivial lemmas which was not proved by Tarski and his co-authors although they are used implicitly. Then we describe Hilbert’s axiom system and its formalization in Coq. Finally, we describe how we can define the concepts of Hilbert’s axiom system and prove the axioms within Tarski’s system.

2 Tarski’s Geometry

2.1 Tarski’s Axiom System

Alfred Tarski worked on the axiomatization and meta-mathematics of Euclidean geometry from 1926, until his death in 1983. Several axiom systems were produced by Tarski and his students. In this section we describe the axiom system we used in the formalization. Further discussion about the history of this axiom system and the different versions can be found in [24]. The axioms can be expressed using first order logic and two predicates. Note that the original theory of Tarski assumes first order logic. Our formalization is performed in a higher order logic setting (the calculus of constructions), hence, the language allowed in the statements and proofs makes the theory more expressible. The meta-theoretical results of Tarski may not apply to our formalization.

betweenness The ternary *betweenness* predicate $\beta A B C$ informally states that B lies on the line AC between A and C .

equidistance The quaternary *equidistance* predicate $AB \equiv CD$ informally means that the distance from A to B is equal to the distance from C to D .

Note that in Tarski’s geometry, only points are primitive objects. In particular, lines are *defined* by two distinct points whereas in Hilbert’s axiom system lines and planes are *primitive objects*. Figure 1 provides the list of axioms that we used in our formalization.

The formalization of this axiom system in Coq is straightforward (Fig. 2). We use the Coq type class mechanism [25] to capture the axiom system. Internally the type class system is based on records containing types, functions and properties about them. Note that we know that this system of axioms has a model: Tuan Minh Pham has shown that these axioms can be derived from Guillhot’s development using an axiom system based on mass points [13].

Identity $\beta A B A \Rightarrow (A = B)$
 Pseudo-Transitivity $AB \equiv CD \wedge AB \equiv EF \Rightarrow CD \equiv EF$
 Symmetry $AB \equiv BA$
 Identity $AB \equiv CC \Rightarrow A = B$
 Pasch $\beta APC \wedge \beta BQC \Rightarrow \exists X, \beta PXB \wedge \beta QXA$
 Euclid $\exists XY, \beta ADT \wedge \beta BDC \wedge A \neq D \Rightarrow$
 $\beta ABX \wedge \beta ACY \wedge \beta XTY$
 $AB \equiv A'B' \wedge BC \equiv B'C' \wedge$
 5 segments $AD \equiv A'D' \wedge BD \equiv B'D' \wedge$
 $\beta ABC \wedge \beta A'B'C' \wedge A \neq B \Rightarrow CD \equiv C'D'$
 Construction $\exists E, \beta ABE \wedge BE \equiv CD$
 Lower Dimension $\exists ABC, \neg \beta ABC \wedge \neg \beta BCA \wedge \neg \beta CAB$
 Upper Dimension $AP \equiv AQ \wedge BP \equiv BQ \wedge CP \equiv CQ \wedge P \neq Q$
 $\Rightarrow \beta ABC \vee \beta BCA \vee \beta CAB$
 Continuity $\forall XY, (\exists A, (\forall xy, x \in X \wedge y \in Y \Rightarrow \beta Axy)) \Rightarrow$
 $\exists B, (\forall xy, x \in X \Rightarrow y \in Y \Rightarrow \beta xBy).$

Fig. 1. Tarski's axiom system.

```

Class Tarski := {
  Tpoint : Type;
  Bet : Tpoint -> Tpoint -> Prop;
  Cong : Tpoint -> Tpoint -> Tpoint -> Prop;
  between_identity : forall A B, Bet A B A -> A=B;
  cong_pseudo_reflexivity : forall A B : Tpoint, Cong A B B A;
  cong_identity : forall A B C : Tpoint, Cong A B C C -> A = B;
  cong_inner_transitivity : forall A B C D E F : Tpoint,
    Cong A B C D -> Cong A B E F -> Cong C D E F;
  inner_pasch : forall A B C P Q : Tpoint,
    Bet A P C -> Bet B Q C -> exists x, Bet P x B /\ Bet Q x A;
  euclid : forall A B C D T : Tpoint,
    Bet A D T -> Bet B D C -> A<>D ->
    exists x, exists y, Bet A B x /\ Bet A C y /\ Bet x T y;
  five_segments : forall A A' B B' C C' D D' : Tpoint,
    Cong A B A' B' -> Cong B C B' C' -> Cong A D A' D' -> Cong B D B' D' ->
    Bet A B C -> Bet A' B' C' -> A <> B -> Cong C D C' D';
  segment_construction : forall A B C D : Tpoint,
    exists E : Tpoint, Bet A B E /\ Cong B E C D;
  lower_dim : exists A, exists B, exists C, ~ (Bet A B C \/ Bet B C A \/ Bet C A B);
  upper_dim : forall A B C P Q : Tpoint,
    P <> Q -> Cong A P A Q -> Cong B P B Q -> Cong C P C Q ->
    (Bet A B C \/ Bet B C A \/ Bet C A B)
}

```

Fig. 2. Tarski's axiom system as a Coq type class.

2.2 Overview of the Formalization of the Book

The formalization closely follows the book [8]. But many lemmas are used implicitly in the proofs and are not stated by the original authors. We first give a quick overview of the different notions introduced in the formal development. Then we provide as an example a proof of a lemma which was not given by the original authors. This lemma is not needed to derive Hilbert's axioms but it is a key lemma for the part of our library about angles. The proof of this lemma represents roughly 100 lines of the 24000 lines of proof of the whole Coq development.

The different concepts involved in Tarski's geometry We followed closely the order given by Tarski to introduce the different concepts of geometry and their associated lemmas. We provide some statistics about the different chapters in Table 1.

Chapter 2: betweenness properties

Chapter 3: congruence properties

Chapter 4: properties of betweenness and congruence This chapter introduces the definition of the concept of collinearity:

Definition 1 (collinearity). *To assert that three points A , B and C are collinear we note: $Col\ A\ B\ C$*

$$Col\ A\ B\ C := \beta\ A\ B\ C \vee \beta\ A\ C\ B \vee \beta\ B\ A\ C$$

Chapter 5: order relation over pair of points The relation `bet_le` between two pair of points formalizes the fact that the distance of the first pair of points is less than the distance between the second pair of points:

Definition 2 (`bet_le`).

$$bet_le\ A\ B\ C\ D := \exists y, \beta\ C\ y\ D \wedge AB \equiv Cy$$

Chapter 6: the ternary relation `out` `Out\ A\ B\ C` means that A , B and C lies on the same line, but A is not between B and C :

Definition 3 (`out`).

$$Out\ P\ A\ B := A \neq P \wedge B \neq P \wedge (\beta\ P\ A\ B \vee \beta\ P\ B\ A)$$

Chapter 7: property of the midpoint This chapter provides a definition for midpoint but the existence of the midpoint will be proved only in Chapter 8.


Definition 4 (midpoint).

$$is_midpoint\ M\ A\ B := \beta\ A\ M\ B \wedge AM \equiv BM$$

Chapter 8: orthogonality lemmas To work on orthogonality, Tarski introduces three relations:

Definition 5 (Per).

$$\text{Per } A B C := \exists C', \text{midpoint } B C C' \wedge AC \equiv AC'$$



Definition 6 (Perp.in).

$$\begin{aligned} \text{Perp.in } X A B C D := & A \neq B \wedge C \neq D \wedge \text{Col } X A B \wedge \text{Col } X C D \wedge \\ & (\forall U V, \text{Col } U A B \Rightarrow \text{Col } V C D \Rightarrow \text{Per } U X V) \end{aligned}$$

Finally, the relation Perp which we note \perp :


Definition 7 (Perp).

$$AB \perp CD := \exists X, \text{Perp.in } X A B C D$$

Chapter 9: position of two points relatively to a line In this chapter, Tarski introduces two predicates to assert the fact that two points which do not belong to a line are either on the same side, or on both sides of the line.


Definition 8 (both sides). Given a line l defined by two distinct points A and B , two points X and Y not on l , are on both sides of l is written: $A \overset{X}{\underset{Y}{-}} B$

$$A \overset{X}{\underset{Y}{-}} B := \exists T, \text{Col } A B T \wedge \beta X T Y$$



Definition 9 (same side). Given a line l defined by two distinct points A and B . Two points X and Y not on l , are on the same side of l is written: $A \overset{X Y}{-} B$


$$A \overset{X Y}{-} B := \exists Z, A \overset{X}{\underset{Z}{-}} B \wedge A \overset{Y}{\underset{Z}{-}} B$$



Chapter 10: orthogonal symmetry The predicate `is_image` allows us to assert that two points are symmetric. Given a line l defined by two distinct points A and B . Two points P and P' are symmetric points relatively to the line l means:

Definition 10 (is_image).

$$is_image\ P\ P'\ A\ B := (\exists X, midpoint\ X\ P\ P' \wedge Col\ A\ B\ X) \wedge (AB \perp PP' \vee P = P')$$




Chapter 11: properties about angles In this chapter, Tarski gives a definition on angle congruence using the similarity of triangles:

Definition 11 (angle congruence).

$$\angle ABC \cong \angle DEF := A \neq B \Rightarrow B \neq C \Rightarrow D \neq E \Rightarrow F \neq F' \Rightarrow$$

$$\exists A', \exists C', \exists D', \exists F' \left\{ \begin{array}{l} \beta\ B\ A\ A' \wedge AA' \equiv ED \wedge \\ \beta\ B\ C\ C' \wedge CC' \equiv EF \wedge \\ \beta\ E\ D\ D' \wedge DD' \equiv BA \wedge \\ \beta\ E\ F\ F' \wedge FF' \equiv BC \wedge \\ A'C' \equiv D'F' \end{array} \right.$$



Definition 12 (in angle).

$$P\ in\ \angle ABC := A \neq B \wedge C \neq B \wedge P \neq B \wedge \exists X, \beta\ A\ X\ C \wedge (X = B \vee Out\ B\ X\ P)$$

Definition 13 (angle comparison).

$$\angle ABC \leq \angle DEF := \exists P, P\ in\ \angle DEF \wedge \angle ABC \cong \angle DEP$$

Chapter 12: parallelism Tarski defines a strict parallelism over two pairs of points:

Definition 14 (parallelism).

$$AB \parallel CD := A \neq B \wedge C \neq D \wedge \neg \exists X, Col\ X\ A\ B \wedge Col\ X\ C\ D$$

Chapter	Number of lemmas	Number of lines of specification	Number of lines of proof
Betweenness properties	16	69	111
Congruence properties	16	54	116
Properties of betweenness and congruence	19	151	183
Order relation over pair of points	17	88	340
The ternary relation out	22	103	426
Property of the midpoint	21	101	758
Orthogonality lemmas	77	191	2412
Position of two points relatively to a line	37	145	2333
Orthogonal symmetry	44	173	2712
Properties about angles	187	433	10612
Parallelism	68	163	3560

Table 1. Statistics about the development.

A Proof Example In this section we give an example of a proof. In [8], Tarski and his co-authors proves that given two angles, one is less or equal to the other one:

Theorem 1 (*lea_cases*).

$$\begin{aligned} \forall ABCDEF, A \neq B \Rightarrow C \neq B \Rightarrow D \neq E \Rightarrow F \neq E \\ \Rightarrow \angle ABC \leq \angle DEF \vee \angle DEF \leq \angle ABC \end{aligned}$$

To prove the lemma *lea_cases*, Tarski uses implicitly the fact that given a line l , two points not on l , are either on the same side of l or on both sides. But he does not give explicitly a proof of this fact. Tarski proved that if two points are on both sides of a line, they are not on the same side (lemma *l9_9*), and if two points are on the same side, they are not on both sides (lemma *l9_9_bis*).

To prove that two points are either on the same side of a line, or on both sides, we need to show that if two points are not on both sides of a line they are on the same side which is the reciprocal lemma of *l9_9_bis*.

We will show the main steps necessary to prove that two points not on a given line l and not on both sides of l are on the same side:

Lemma (*not_two_sides_one_side*).

$$\neg Col ABX \Rightarrow \neg Col ABY \Rightarrow \neg A \frac{X}{Y} B \Rightarrow A \frac{X}{Y} B$$

Proof. The lemmas used in this proof are shown on Table 2.

Step one:

First we build the point P_X on the line AB such that $XP_X \perp AB$. The existence of P_X is proved by the lemma *l8_18_existence* (the lemmas used in this proof are provided in Table 2).

Lemma 1 (l8-21).

$$\forall ABC, A \neq B \Rightarrow \exists P, \exists T, AB \perp PA \wedge Col ABT \wedge \beta CTP$$

Lemma (or_bet_out).

$$\forall ABC, A \neq B \Rightarrow C \neq B \Rightarrow \beta ABC \vee Out BAC \vee \neg Col ABC$$

Lemma (l8-18-existence³).

$$\forall ABC, \neg Col ABC \Rightarrow \exists X, Col ABX \wedge AB \perp CX$$

Lemma (perp_perp_col).

$$\forall AB XY P, P \neq A \Rightarrow Col ABP \Rightarrow AB \perp XP \Rightarrow PA \perp YP \Rightarrow Col YXP$$

Lemma (out_one_side).

$$\forall ABXY, (\neg Col ABX \vee \neg Col ABY) \Rightarrow Out AXY \Rightarrow A \overline{XY} B$$

Lemma (l8-8-2).

$$\forall PQABC, P \overline{AC} Q \Rightarrow P \overline{AB} Q \Rightarrow P \overline{BC} Q$$

Table 2. Lemmas used in the proof.

Step two:


To prove that the points X and Y are on the same side of the line AB we prove the existence of a point P verifying $A \overline{XP} B \wedge A \overline{YP} B$ as required by the definition of the relation “same side” (Definition 9).

The key step of the proof is the lemma *l8-21* which allows to build such a point P . Then we will establish that this point P verifies the expected property.

To use the lemma *l8-21* we need a point on the line AB different from P_X . Since $A \neq B$, the point P_X must be different from A or B . For our proof we suppose that $P_X \neq A$. The same proof could be done using B instead of A .

Thus we can instantiate the lemma *l8-21* with the points P_X , A and Y :

$$P_X \neq A \Rightarrow \exists P, \exists T, P_X A \perp PP_X \wedge Col P_X AT \wedge \beta YTP$$



Step three:

We can trivially prove that Y and P are located on both sides of the line AB since T is collinear with A and Px . Therefore T is collinear with A and B , and T is between P and Y which correspond exactly to the definition of the “both sides” relation (Definition 8). Thus we get:

$$A \underset{Y}{\overset{P}{-}} B \quad (1)$$

Step four:

Now it remains to show that X and P are located on both sides of the line AB . First, we prove that X , Px and P are collinear using the lemma *perp_perp_col*.

Second, we use the lemma *or_bet_out* applied to the three points X , P_X and P to distinguish three cases:

1. $\beta X P_X P$
2. $Out P_X X P$
3. $\neg Col X P_X P$

1. The first case gives trivially a proof of $A \underset{P}{\overset{X}{-}} B$ since $\beta X P_X P$ and $Col A B P_X$ which is the definition of the relation “both sides” (Definition 8). Since $A \underset{P}{\overset{X}{-}} B$ and $A \underset{P}{\overset{Y}{-}} B$ (step 3) we can conclude $A \underset{XY}{-} B$ using the definition of the relation “same side” (Definition 9).

2. The second case also leads to a contradiction:

The lemma *out_one_side* allows to deduce $A \underset{PX}{-} B$.

Using *out_one_side* applied to $P_X A X P$ we have:

$$(\neg Col P_X A X \vee \neg Col P_X A P) \Rightarrow Out P_X X P \Rightarrow P_X \underset{XP}{-} A$$

Since P_X is collinear with A and B we also get:

$$A \underset{XP}{-} B \iff A \underset{PX}{-} B \quad (\text{symmetry of “same side”}) \quad (2)$$

Finally, we will derive the contradiction using lemma *l8_8_2*:

Using *l8_8_2* applied to A , B , P , X and Y , we get:

$$\underbrace{A \underset{Y}{\overset{P}{-}} B}_{(1)} \Rightarrow \underbrace{A \underset{PX}{-} B}_{(2)} \Rightarrow A \underset{Y}{\overset{X}{-}} B$$

The hypothesis $\neg A \underset{Y}{\overset{X}{-}} B$ is in contradiction with the conclusion $A \underset{Y}{\overset{X}{-}} B$.

3. The third case leads easily to a contradiction since we proved $Col X P_X P$. \square

3 Hilbert’s Axiom System

Hilbert’s axiom system is based on two abstract types: points and lines (as we limit ourselves to 2-dimensional geometry we do not introduce ‘planes’ and the related axioms). In Coq’s syntax we have:

```
Point : Type
Line  : Type
```

We assume that the type `Line` is equipped with an equivalence relation `EqL` which denotes equality between lines:

```
EqL    : Line -> Line -> Prop
EqL_Equiv : Equivalence EqL
```

We do not use Leibniz equality (the built-in equality of Coq), because when we will define the notion of line inside Tarski's system, the equality will be a defined notion. Note that we do not follow closely the Hilbert's presentation because we use an explicit definition of the equality relation.

We assume that we have a relation of incidence between points and lines:

```
Incid : Point -> Line -> Prop
```

We also assume that we have a relation of betweenness:

```
BetH : Point -> Point -> Point -> Prop
```

Notice that contrary to the `Bet` relation of Tarski, the one of Hilbert implies that the points are distinct.

The axioms are classified by Hilbert into five groups: Incidence, Order, Parallel, Congruence and Continuity. We formalize here only the first four groups, leaving out the continuity axiom. We provide details only when the formalization is not straightforward.

3.1 Incidence Axioms

Axiom (I 1). *For every two distinct points A, B there exists a line l such that A and B are incident to l .*

```
line_existence : forall A B, A <> B -> exists l, Incid A l /\ Incid B l;
```

Axiom (I 2). *For every two distinct points A, B there exists at most one line l such that A and B are incident to l .*

```
line_unicity : forall A B l m, A <> B ->
  Incid A l -> Incid B l -> Incid A m -> Incid B m -> EqL l m;
```

Axiom (I 3). *There exists at least two points on a line. There exists at least three points that do not lie on a line.*

```
two_points_on_line : forall l, exists A, exists B,
  Incid B l /\ Incid A l /\ A <> B
```

```
ColH A B C := exists l, Incid A l /\ Incid B l /\ Incid C l
```

```
plan : exists A, exists B, exists C, ~ ColH A B C
```

3.2 Order Axioms

It is straightforward to formalize the axioms of order:

Axiom (II 1). *If a point B lies between a point A and a point C then the point A, B, C are three distinct points through a line, and B also lies between C and A .*

```
between_col : forall A B C : Point, BetH A B C -> ColH A B C
between_comm : forall A B C : Point, BetH A B C -> BetH C B A
```

Axiom (II 2). *For two distinct points A and B , there always exist at least one point C on line AB such that B lies between A and C .*


```
between_out : forall A B : Point,
  A <> B -> exists C : Point, BetH A B C
```

Axiom (II 3). *Of any three distinct points situated on a straight line, there is always one and only one which lies between the other two.*

```
between_only_one : forall A B C : Point,
  BetH A B C -> ~ BetH B C A /\ ~ BetH B A C
```

```
between_one : forall A B C, A <> B -> A <> C -> B <> C -> ColH A B C ->
  BetH A B C \/ BetH B C A \/ BetH B A C
```

Axiom (II 4 - Pasch). *Let A, B and C be three points that do not lie in a line and let a be a line (in the plane ABC) which does not meet any of the points A, B, C . If the line a passes through a point of the segment AB , it also passes through a point of the segment AC or through a point of the segment BC .*



To give a formal definition for this axiom we need an extra definition:

```
cut l A B := ~Incid A l /\ ~Incid B l /\
  exists I, Incid I l /\ BetH A I B
```

```
pasch : forall A B C l, ~ColH A B C -> ~Incid C l -> cut l A B ->
  cut l A C \/ cut l B C
```

3.3 Parallel Axiom

As we are in a two-dimensional setting, we follow Hilbert and say that two lines are parallel when they have no point in common. Then Euclid's axiom states that there exists a unique line parallel to another line l passing through a given point P . Note that as the notion of parallel is strict we need to assume that P does not belong to l .

```

Para l m := ~ exists X, Incid X l /\ Incid X m;
euclid_existence : forall l P, ~ Incid P l -> exists m, Para l m;
euclid_unicity : forall l P m1 m2, ~ Incid P l ->
  Para l m1 -> Incid P m1 ->
  Para l m2 -> Incid P m2 -> EqL m1 m2;

```

3.4 Congruence Axioms

The congruence axioms are the most difficult to formalize because Hilbert does not provide clear definitions for all the concepts occurring in the axioms. Here is the first axiom:

***Axiom (IV 1).** If A, B are two points on a straight line a , and if A' is a point upon the same or another straight line a' , then, upon a given side of A' on the straight line a' , we can always find one and only one point B' so that the segment AB is congruent to the segment $A'B'$. We indicate this relation by writing $AB \equiv A'B'$.*

To formalize the notion of “on a given side”, we split the axiom into two parts: existence and uniqueness. We state the existence of a point on each side, and we state the uniqueness of this pair of points.

```

cong_existence : forall A B l M, A <> B -> Incid M l ->
  exists A', exists B',
  Incid A' l /\ Incid B' l /\ BetH A' M B' /\
  CongH M A' A B /\ CongH M B' A B

cong_unicity : forall A B l M A' B' A'' B'', A <> B -> Incid M l ->
  Incid A' l -> Incid B' l ->
  BetH A' M B' -> CongH M A' A B -> CongH M B' A B ->
  Incid A'' l -> Incid B'' l ->
  BetH A'' M B'' -> CongH M A'' A B -> CongH M B'' A B ->
  (A' = A'' /\ B' = B'') \/ (A' = B'' /\ B' = A'')

```

***Axiom (IV 2).** If a segment AB is congruent to the segment $A'B'$ and also to the segment $A''B''$, then the segment $A'B'$ is congruent to the segment $A''B''$.*

The formalization of this axiom is straightforward:

```

cong_pseudo_transitivity : forall A B A' B' A'' B'',
  CongH A B A' B' -> CongH A B A'' B'' -> CongH A' B' A'' B''

```

Note that from the last two axioms we can deduce the reflexivity of the relation \equiv .

Axiom (IV 3). *Let AB and BC be two segments of a straight line a which have no points in common aside from the point B , and, furthermore, let $A'B'$ and $B'C'$ be two segments of the same or of another straight line a' having, likewise, no point other than B' in common. Then, if $AB \equiv A'B'$ and $BC \equiv B'C'$, we have $AC \equiv A'C'$.*

First, we define when two segments have no common points. Note that we do not introduce a type of segments for the sake of simplicity.

Definition disjoint A B C D :=
 \sim exists P, Between_H A P B /\ Between_H C P D.

Then, we can formalize the axioms IV 3:

```
addition: forall A B C A' B' C',
  ColH A B C -> ColH A' B' C' ->
  disjoint A B B C -> disjoint A' B' B' C' ->
  CongH A B A' B' -> CongH B C B' C' -> CongH A C A' C'
```

Angle Hilbert defines an angle with two distinct half-lines emanating from a same point. The imposed condition that two half-lines be distinct excludes the null angle from the definition. Tarski defines an angle with three points. Two of them have to be different from the third which is the top of the angle. Such a definition allows null angles. For our formalization of Hilbert, we choose to differ slightly from his definition and use a triple of points. Our definition includes the null angle. Defining angles using half-lines consists in a definition involving four points and the proof that two of them are equal. It is just simpler to use only three points.

```
Record Triple {A:Type} : Type :=
  build_triple {V1   : A ;
                V    : A ;
                V2   : A ;
                Pred : V1 <> V /\ V2 <> V}.
```

Definition angle := build_triple Point.

Axiom (IV-4). *Given an angle α , an half-line h emanating from a point O and given a point P , not on the line generated by h , there is a unique half-line h' emanating from O , such that the angle α' defined by (h, O, h') is congruent with α and such that every point inside α' and P are on the same side relatively to the line generated by h .*

To formalize this axiom we need definitions for the underlying concepts. Hilbert uses the “same side” notion to define interior points of an angle:

Given two half-lines h and h' emanating from a same point, every point P on the same side of h as a point of h' and on the same side of h' as a point of h is in the interior of the angle defined by h and h' .

Hilbert gives a formal definition of the relative position of two points of a line compared to a third point:

$$\forall A O B, \beta_H(A, O, B) \iff A \text{ and } B \text{ are on both sides of } O \quad (3)$$

$$\forall A A' O, \beta_H(A, A', O) \vee \beta_H(A', A, O) \iff A \text{ and } A' \text{ are on the same side of } O \quad (4)$$



This second definition (4) allows to define the notion of half-line: given a line l , and a point O on l , all pairs of points laying on the same side of O belong to the same half-line emanating from O .

`outH P A B := BetH P A B \/\ BetH P B A \/\ (P <> A /\ A = B);`

We define the interior of an angle as following:

```
InAngleH a P :=
  (exists M, BetH (V1 a) M (V2 a) /\
   ((outH (V a) M P) \/\ M = (V a))) \/\
   outH (V a) (V1 a) P \/\
   outH (V a) (V2 a) P;
```

Hilbert gives a formal definition of the relative position of two points and a line:

`same_side A B l := exists P, cut l A P /\ cut l B P;`

Then the fourth axiom is a little bit verbose because we need to manipulate the non-degeneracy conditions for the existence of the angles and half-lines.

```
aux : forall (h h1 : Hline), P1 h = P1 h1 -> P2 h1 <> P1 h;
hcong_4_existence: forall a h P,
  ~Incid P (line_of_hline h) -> ~ BetH (V1 a)(V a)(V2 a) ->
  exists h1, (P1 h) = (P1 h1) /\ (forall CondAux : P1 h = P1 h1,
  CongaH a (angle (P2 h) (P1 h) (P2 h1) (conj (sym_not_equal (Cond h))
  (aux h h1 CondAux))) /\
  (forall M, ~ Incid M (line_of_hline h) /\ InAngleH (angle (P2 h) (P1 h) (P2 h1)
  (conj (sym_not_equal (Cond h)) (aux h h1 CondAux))) M ->
  same_side P M (line_of_hline h));
```

The uniqueness axiom requires an equality relation between half-lines⁴:

⁴ P1 is the function to access to the first point of the half-line and P2 the second point.

```

hEq : relation Hline := fun h1 h2 => (P1 h1) = (P1 h2) /\
  ((P2 h1) = (P2 h2) \/ BetH (P1 h1) (P2 h2) (P2 h1) \/
   BetH (P1 h1) (P2 h1) (P2 h2));

hline_construction a (h: Hline) P (hc:Hline) H :=
(P1 h) = (P1 hc) /\
CongaH a (angle (P2 h) (P1 h) (P2 hc) (conj (sym_not_equal (Cond h)) H)) /\
(forall M, InAngleH (angle (P2 h) (P1 h) (P2 hc)
  (conj (sym_not_equal (Cond h)) H)) M ->
  same_side P M (line_of_hline h));

hcong_4_unicity : forall a h P h1 h2 HH1 HH2,
  ~Incid P (line_of_hline h) -> ~ BetH (V1 a)(V a)(V2 a) ->
  hline_construction a h P h1 HH1 -> hline_construction a h P h2 HH2 ->
  hEq h1 h2

```

The last axiom is easier to formalize as we already have all the required definitions:

***Axiom (IV 5).** If the following congruences hold $AB \equiv A'B'$, $AC \equiv A'C'$, $\angle BAC \equiv \angle B'A'C'$ then $\angle ABC \equiv \angle A'B'C'$*

```

cong_5 : forall A B C A' B' C',
  forall H1 : (B<>A /\ C<>A),
  forall H2 : (B'<>A' /\ C'<>A'),
  forall H3 : (A<>B /\ C<>B),
  forall H4 : (A'<>B' /\ C'<>B'),
  CongH A B A' B' -> CongH A C A' C' ->
  CongaH (angle B A C H1) (angle B' A' C' H2) ->
  CongaH (angle A B C H3) (angle A' B' C' H4)

```

4 Hilbert follows from Tarski

In this section, we describe the main result of our development, which consists in a formal proof that Hilbert's axioms can be defined and proved within Tarski's axiom system. We prove that Tarski's system constitutes a model of Hilbert's axioms (continuity axioms are excluded from this study).

Encoding the concepts of Hilbert within Tarski's geometry In this section, we describe how we can define the different concepts involved in Hilbert's axiom system using the definition of Tarski. We also compare the definitions in the two systems when they are not equivalent. We will define the concepts of line, betweenness, out, parallel, angle.

Lines: To define the concept of line within Tarski, we need the concept of two distinct points. For our formalization in Coq, we use a dependent type which consists in a record containing two elements of a given type A together with

a proof that they are distinct. We use a polymorphic type instead of defining directly a couple of points for a technical reason. To show that we can instantiate Hilbert type class in the context of Tarski, Coq will require that some definitions in the two corresponding type classes share the definition of this record.

```
Record Couple {A:Type} : Type :=
  build_couple {P1: A ; P2 : A ; Cond: P1 <> P2}.
```

Then, we can define a line by instantiating A with the type of the points to obtain a couple of points:

```
Definition Line := @Couple Tpoint.
Definition Lin := build_couple Tpoint.
```

But, if for example we have four distinct points A , B , C and D which are collinear, the lines AB and CD are different according to Leibniz equality (the standard equality of Coq), hence we need to define our own equality on the type of lines:

```
Definition Eq : relation Line :=
  fun l m => forall X, Incident X l <-> Incident X m.
```

We can easily show that this relation is an equivalence relation. And we also show that it is a proper morphism for the `Incident` predicate.

```
Lemma eq_incident : forall A l m,
  Eq l m -> (Incident A l <-> Incident A m).
```

Betweenness: As noted before, Hilbert's betweenness definition differs from Tarski's one. Hilbert define a strict betweenness which requires that the three points concerned by the relation to be different. With Tarski, this constraint does not appear. Hence we have:

```
Definition Between_H A B C := Bet A B C /\ A <> B /\ B <> C /\ A <> C.
```

Out: Here is a definition of the concept of 'out' defined using the concepts of Hilbert:

```
Definition outH :=
  fun P A B => Between_H P A B \/ Between_H P B A \/ (P <> A /\ A = B).
```

We can show that it is equivalent to the concept of 'out' of Tarski:

```
Lemma outH_out : forall P A B, outH P A B <-> out P A B.
```

Parallels: The concept of parallel lines in Tarski's formalization includes the case where the two lines are equal, whereas it is excluded in Hilbert's. Hence we have:

```
Lemma Para_Par : forall A B C D, forall HAB: A<>B, forall HCD: C<>D,
  Para (Lin A B HAB) (Lin C D HCD) -> Par A B C D
```

where `par` denotes the concept of parallel in Tarski's system and `Para` in Hilbert's. Note that the converse is not true.

Angle: As noted before we define an angle by a triple of points with some side conditions. We use a polymorphic type for the same reason as for the definition of lines:

```
Record Triple {A:Type} : Type :=
  build_triple {V1  : A ;
                V   : A ;
                V2  : A ;
                Pred : V1 <> V /\ V2 <> V}.
```

```
Definition angle := build_triple Tpoint.
```

```
Definition InAngleH a P :=
  (exists M, Between_H (V1 a) M (V2 a) /\ ((outH (V a) M P) \/\ M=(V a)))
  \/\ outH (V a) (V1 a) P \/\ outH (V a) (V2 a) P.
```

```
Lemma in_angle_equiv : forall a P, (P <> (V a) /\ InAngleH a P) <->
  InAngle P (V1 a) (V a) (V2 a).
```

Main result Once the required concepts have been defined, we can use our large set of results describe in Sec. 2.2 to prove every axiom of Hilbert's system. To capture our result within Coq we assume we have an instance `T` of the class `Tarski`, and we show that we have an instance of the class `Hilbert`: This requires 1200 lines of formal proof. From a technical point, to capture this fact in Coq, we could have built a *functor* from a module type to another module type. We chose the approach based on type classes, because type classes are first class citizens in Coq.

```
Section Hilbert_to_Tarski.
```

```
Context '{T:Tarski}.
```

```
Instance Hilbert_follow_from_Tarski : Hilbert.
```

```
Proof.
```

```
... (* omitted here *)
```

```
Qed.
```

```
End Hilbert_to_Tarski.
```

5 Conclusion

We have proposed the first formal proof that Hilbert's axioms can be derived from Tarski's axioms. This work can now serve as foundations for the many other Coq developments about geometry. The advantage of Tarski's axioms lies in the fact that there are few axioms and most of them have been shown to be independent from the others. Moreover a large part of our proofs are independent of some axioms. For instance the axiom of Euclid is used for the first time in Chapter 12.

Hence, the proofs derived before this chapter are also valid in absolute geometry. In the future we plan to reconstruct the foundations of Frédérique Guilhot’s formalization of high-school geometry and of our formalizations of automated deduction methods in geometry [20, 21] using Tarski’s axioms.

Availability

The full Coq development consists of more than 500 lemmas and 24000 lines of formal proof. The formal proofs and definitions with hypertext links and dynamic figures can be found at the following urls:

<http://dpt-info.u-strasbg.fr/~narboux/tarski.html>
<http://gabrielbraun.free.fr/Geometry/Tarski/>

Acknowledgments

We would like to thank our reviewers for their numerous remarks which helped improving this paper.

References

1. Avigad, J., Dean, E., Mumma, J.: A formal system for euclid’s elements. *Review of Symbolic Logic* **2** (2009) 700–768
2. Coq development team: The Coq Proof Assistant Reference Manual, Version 8.3. TypiCal Project. (2010)
3. Dehlinger, C., Dufourd, J.F., Schreck, P.: Higher-Order Intuitionistic Formalization and Proofs in Hilbert’s Elementary Geometry. In: ADG’00. Volume 2061 of LNAI, Springer-Verlag (2000) 306–324
4. Paulson, L.C.: The Isabelle reference manual (2006)
5. Meikle, L., Fleuriot, J.: Formalizing Hilbert’s Grundlagen in Isabelle/Isar. In: TPHOLS’03. Volume 2758 of LNCS. (2003) 319–334
6. Scott, P., Fleuriot, J.: An Investigation of Hilberts Implicit Reasoning through Proof Discovery in Idle-Time. In Schreck, P., Narboux, J., Richter-Gebert, J., eds.: *Automated Deduction in Geometry*. Volume 6877 of Lecture Notes in Computer Science. Springer Berlin / Heidelberg (2011) 182200 10.1007/978-3-642-25070-5_11.
7. Tarski, A., Givant, S.: Tarski’s system of geometry. *The bulletin of Symbolic Logic* **5**(2) (June 1999)
8. Schwabhäuser, W., Szmielew, W., Tarski, A.: *Metamathematische Methoden in der Geometrie*. Springer-Verlag (1983) In german.
9. Quaife, A.: Automated development of Tarski’s geometry. *Journal of Automated Reasoning* **5**(1) (1989) 97–118
10. Gupta, H.N.: Contributions to the axiomatic foundations of geometry. PhD thesis, University of California, Berkley (1965)
11. Narboux, J.: Toward the use of a proof assistant to teach mathematics. In: The Seventh International Conference on Technology in Mathematics Teaching (ICTMT7), Bristol, United Kingdom (2005)

12. Guillhot, F.: Formalisation en Coq et visualisation d'un cours de géométrie pour le lycée. *TSI* **24** (2005) 1113–1138 In french.
13. Pham, T.M.: Description formelle de propriété géométriques. PhD thesis, Université de Nice - Sophia-Antipolis (2011)
14. Pichardie, D., Bertot, Y.: Formalizing convex hulls algorithms. In: *TPHOLs'2001*. Volume 2152 of LNCS., Springer-Verlag (2001) 346–361
15. Brun, C., Dufourd, J.F., Magaud, N.: Designing and Proving Correct a Convex Hull Algorithm with Hypermaps in Coq. *Computational Geometry: Theory and Applications* (2010) To appear.
16. Meikle, L., Fleuriot, J.: Mechanical theorem proving in computation geometry. In Hong, H., Wang, D., eds.: *ADG'04*. Volume 3763 of LNAI., Springer-Verlag (2005) 1–18
17. Chou, S.C., Gao, X.S., Zhang, J.Z.: *Machine Proofs in Geometry*. World Scientific (1994)
18. Narboux, J.: A decision procedure for geometry in Coq. In: *TPHOLs'04*. Volume 3223 of LNCS., Springer-Verlag (2004) 225–240
19. Narboux, J.: Formalisation et automatisé du raisonnement géométrique en Coq. PhD thesis, Université Paris Sud (September 2006) In french.
20. Janicic, P., Narboux, J., Quaresma, P.: The Area Method : a Recapitulation. *Journal of Automated Reasoning* **48**(4) (2012) 489–532
21. Genevaux, J.D., Narboux, J., Schreck, P.: Formalization of Wu's simple method in Coq. In Jouannaud, J.P., Shao, Z., eds.: *CPP 2011 First International Conference on Certified Programs and Proofs*. Volume 7086 of LNCS., Kenting, Taiwan, Province Of China, Springer-Verlag (December 2011) 71–86
22. Chou, S.C.: *Mechanical Geometry Theorem Proving*. D. Reidel Publishing Company (1988)
23. Pham, T.M., Bertot, Y., Narboux, J.: A Coq-based Library for Interactive and Automated Theorem Proving in Plane Geometry. In: *The 11th International Conference on Computational Science and Its Applications (ICCSA 2011)*. Volume 6785 of LNCS., Santander, Spain, Springer-Verlag (2011) 368–383
24. Narboux, J.: Mechanical Theorem Proving in Tarski's geometry. In Eugenio Roanes Lozano, F.B., ed.: *Automated Deduction in Geometry 2006*. Volume 4869 of LNCS., Pontevedra, Spain, Francisco Botana, Springer (2007) 139–156
25. Sozeau, M., Oury, N.: First-class type classes. In Mohamed, O.A., Muñoz, C., Tahar, S., eds.: *TPHOLs*. Volume 5170 of LNCS., Springer (2008) 278–293